



STARLiMS Clean Batch Specifications

Index

- [Document Update History](#)
- [Overview](#)
- [Limitations and Restrictions](#)
- [Servers Setup](#)
- [Batch Job](#)
 - Directories Cleaned
- [Bat Files Summary](#)
 - StartCleanup.bat
 - MainCleanUp.bat
 - StarlimsCleanup.bat
 - iisCleanUp.bat
- [Task Scheduler Configuration](#)
- [Resources and Contacts](#)
- [Appendix](#)
 - Bat Files
 - [StartCleanup.bat](#)
 - [MainCleanUp.bat](#)
 - [StarlimsCleanUp.bat](#)
 - [iisCleanUp.bat](#)



STARLiMS Clean Batch Specifications

Document Update History

Update history for this document

Date	Person	Description
2018.12.07	James Yoder	Addition of Servers Setup section
2018.11.19	James Yoder	Created Document

[Back to Index](#)




STARLiMS Clean Batch Specifications

Overview

The STARLiMS application is a **Laboratory Information Management System (LIMS)** responsible for tracking samples submitted to the Clinical and Environmental labs at ADHS. Functionality performed in this process includes Accessioning, Assigning test, tracking results and generating a final report for clients.

Over time, some directories on the server associated with the STARLiMS application tend to accumulate a high quantity of temporary or one-time use files. The software vendor (Abbott) recommends that these files be periodically purged. Rather relying on a system administrator to manually perform these task we propose creating a batch job to do this and scheduling the batch job to execute off-hours on a regular schedule.

STARLiMS Sample LifeCycle



Limitations and Restrictions

- This clean job will not delete sub directories if being utilized by the STARLiMS application (file lock).
- Retention Days parameter does not apply to sub-folders or content in those sub folders in the directories that have been identified to have sub folders cleaned.

[Back to Index](#)



STARLiMS Clean Batch Specifications

Servers Setup

Two application servers are associated with the STARLiMS application. Each server will contain its own iteration of the Clean Batch job on it. Each batch job is responsible for cleaning the associated directories for each instance of the STARLiMS application residing on the server. Provided below is a graphical representation of each server along with a list of each of the instances of STARLiMS residing on each server.

As can be seen below, the DEVTEST server (hsphxmstarlims2) contains 4 instances of STARLiMS, while the PROD server (hsphxmstarlims1) contains two instances of the STARLiMS application.



[Back to Index](#)




STARLiMS Clean Batch Specifications

Batch Job

The batch job ADHS_STARLiMS_Clean has been configured in the Task Scheduler to kick off the MainCleanUp.bat script responsible for cleaning the STARLiMS associated directories. On completion a log file (**CleanUp.log**) should be generated in the **C:\STARLiMS\AdhsBatch** directory.

STARLiMS Clean Batch Job



The following directories are cleaned in this job:

- Print Service Directories
 - C:\STARLiMS\ReportPrintingService\Done
 - C:\STARLiMS\ReportPrintingService>Error
 - C:\STARLiMS\ReportPrintingService_Dest
- Log Directories
 - C:\STARLiMS\%AppPoolName%\Log
 - C:\STARLiMS\%AppPoolName%\Log\Users
- Reports Cache and sub-directories
 - C:\STARLiMS\%AppPoolName%\WorkPath\ReportsCache
- Shadow Cache and sub-directories
 - C:\STARLiMS\%AppPoolName%\ShadowCache
- Work Path Temp and sub-directories
 - C:\STARLiMS\%AppPoolName%\WorkPath\Temp
- IIS Application Pool and sub-directories
 - C:\STARLiMS\%AppPoolName%\bin\LocalCache

WHERE %AppPoolDir% represents the STARLiMS Instance Application Pool Name.

[Back to Index](#)



STARLiMS Clean Batch Specifications

Bat Files Summary

There are four BAT files associated with the execution of this job. Provided below is a brief summary of each of these files. Detailed code for each file is located in the [Appendix](#).

File Name	Parameters	Description
StartCleanup.bat	None	This file is responsible for calling the MainCleanup.bat file and directing all output to the CleanUp.log file. MainCleanup.bat
MainCleanup.bat	<p>Environment – Indicates what environment is to be cleaned. All STARLiMS instances of the specified environment will be cleaned. Valid values include:</p> <ul style="list-style-type: none"> • DevTest • PROD <p>TestMode – Used to indicate if the BAT file should go in Debug mode. Valid values include:</p> <ul style="list-style-type: none"> • CleanHouse • Debug <p>RetentionDays – The number of days a file is to be retained before being deleted. This value must be an integer.</p>	Driving program for performing the cleanup. This file calls the supporting bat files as needed to complete the cleanup.
StarlimsCleanup.bat	<p>Instance – STARLiMS instance to be cleaned valid values include:</p> <ul style="list-style-type: none"> • ClinicalDEV • ClinicalTEST • ClinicalPROD • ChemistryDEV • ChemistryTEST • ChemistryPROD <p>TestMode – Used to indicate if the BAT file should go in Debug mode. Valid values include:</p> <ul style="list-style-type: none"> • CleanHouse • Debug <p>RetentionDays – The number of days a file is to be retained before being deleted. This value must be an integer.</p>	Cleans up directories residing in the STARLiMS application instance.



STARLiMS Clean Batch Specifications

File Name	Parameters	Description
iisCleanUp.bat	<p>Instance – STARLiMS instance to be cleaned valid values include:</p> <ul style="list-style-type: none">• <code>ClinicalDEV</code>• <code>ClinicalTEST</code>• <code>ClinicalPROD</code>• <code>ChemistryDEV</code>• <code>ChemistryTEST</code>• <code>ChemistryPROD</code> <p>TestMode – Used to indicate if the BAT file should go in Debug mode. Valid values include:</p> <ul style="list-style-type: none">• <code>CleanHouse</code>• <code>Debug</code>	Cleanup sub directories


[Back to Index](#)



STARLiMS Clean Batch Specifications

Task Scheduler Configuration

The Task Scheduler is a windows application used for scheduling task. We will configure this application to schedule the STARLiMS Batch job (**ADHS_STARLiMS_Clean**).



Configuration Details

- General tab
 - Name: **ADHS_STARLiMS_Clean**
 - Location: **\ADHS_Batch**
 - Description: **This batch file will clean up directories associated with the STARLiMS application, it will be applied to all instances on this server (ClinicalDEV, ClinicalTEST, ChemistryDEV, ChemistryTEST).**
 - Security Options
 - When running use the following user account: **HS\Starlims10_SVC**
 - Run Whether user is logged in or not: **Selected**
 - Run with highest privileges: **Checked**
 - Triggers tab
 - Trigger: **Weekly**
 - Details: **At 2:30 AM every Monday**
 - Status: **Enabled**
 - Actions tab
 - Action: **Start a Program**
 - Settings:
 - Program/Script: **C:\STARLiMS\AdhsBatch\StartCleanup.bat**
 - Start in (optional): **C:\STARLiMS\AdhsBatch**

[Back to Index](#)



STARLiMS Clean Batch Specifications

Resources and Contacts

Contacts

Role	Contact	Contact Info	Comments
Server Administrator	Sandhya Kaveti	Email: sandhya.kaveti@azdhs.gov Phone: (602) 542-1159	
LIMS Administrator	James Yoder	Email: james.yoder@azdhs.gov Phone: (602) 542-5787	
LIMS Admin Backup	Rupal Mehta	Email: Rupal.Mehta@azdhs.gov Phone: (602) 364-1884	

Resources

Type	Details	Comments
Service Account	Account: hs\Starlims10_SVC	This account used in the Task scheduler to run the cleanup job.
DEV/TEST Server	hsphxmstarlims2.hs.azdhs.gov	The STARLiMS application server host the Clinical and Chemistry DEV/TEST instances.
PROD Server	Hsphxmstarlims1.hs.azdhs.gov	The STARLiMS application server host the Clinical and Chemistry PROD instances
Clinical DEV Instance	STARLIMS10.Clinical107-DEV	This is the Application Pool name and the sub-directory name in the C:\STARLIMS directory used to host this instance of STARLiMS
Clinical TEST Instance	STARLIMS10.Test	This is the Application Pool name and the sub-directory name in the C:\STARLIMS directory used to host this instance of STARLiMS
Clinical PROD Instance	STARLIMS10.ClinicalLims.azdhs.gov	This is the Application Pool name and the sub-directory name in the C:\STARLIMS directory used to host this instance of STARLiMS
Chemistry DEV Instance	STARLIMS10.ChemLims-Dev.azdhs.gov	This is the Application Pool name and the sub-directory name in the C:\STARLIMS directory used to host this instance of STARLiMS
Chemistry TEST Instance	STARLIMS10.ChemLims-Test.azdhs.gov	This is the Application Pool name and the sub-directory name in the C:\STARLIMS directory used to host this instance of STARLiMS
Chemistry PROD Instance	STARLIMS10.ChemLims.azdhs.gov	This is the Application Pool name and the sub-directory name in the C:\STARLIMS directory used to host this instance of STARLiMS

[Back to Index](#)



Appendix



STARLiMS Clean Batch Specifications

StartCleanup.bat

```
CALL MainCleanUp.bat DevTest CleanHouse 14 >CleanUp.log 2>&1
```

[Back to Index](#)



STARLiMS Clean Batch Specifications

MainCleanUp.bat

```
ECHO OFF
REM: -----MainCleanUp-----
REM:---DESCRIPTION---
REM: This is the driver script used to direct the STARLiMS directories and
REM: IIS files cleanup
REM: The following BAT files are used to accomplish this:
REM: 1. StarlimsCleanUp.bat
REM: 2. iisCleanUp.bat
REM:--- ---

REM:---PARAMETERS---
REM: 1. Environment
REM: DevTest, PROD
REM: 2. TestMode
REM: CleanHouse, Debug
REM: 3. RetentionDays
REM:--- ---

REM:---HISTORY---
REM: 2018.11.16 JTY: Create bat file
REM:--- ---
REM: -----
REM: -----

REM:-----Script Setup-----
CLS
SET Environment="%~1"
SET TestMode="%~2"
SET RetentionDays=-%3
IF %RetentionDays% gtr 0 (
    SET RetentionDays=-%RetentionDays%
)
SET dirBatch=%~dp0
ECHO -----MainCleanUp-----
ECHO ---StartUp---
ECHO Date: %date%
ECHO Time: %time%
ECHO Environment = %Environment%
ECHO TestMode = %TestMode%
ECHO RetentionDays = %RetentionDays%
ECHO --- ---
IF %Environment%==" " GOTO NothingToDo
REM:-----

REM:-----TestMode-----
IF %TestMode%=="Debug" (
    ECHO ON
    SET cmdAction=cmd /c echo Kill @file
)
IF %TestMode%=="CleanHouse" (SET cmdAction=cmd /c del @path /Q)
REM:-----

REM:-----PrintServices-----
ECHO ---PrintServices---
```

Continued on Next Page



STARLiMS Clean Batch Specifications

```
SET FileType=*
REM:--Clean Done Dir--
SET dirTarget=C:\STARLiMS\ReportPrintingService\Done
ECHO    Clean [FileType %FileType%] [Age %RetentionDays%] [DirPath %dirTarget%] --
FORFILES /P %dirTarget% /M %FileType% /S /C "%cmdAction%" /D %RetentionDays%
REM:--Clean Error Dir--
SET dirTarget=C:\STARLiMS\ReportPrintingService\Error
ECHO    Clean [FileType %FileType%] [Age %RetentionDays%] [DirPath %dirTarget%] --
FORFILES /P %dirTarget% /M %FileType% /S /C "%cmdAction%" /D %RetentionDays%
REM:--Clean _Dest Dir--
SET dirTarget=C:\STARLiMS\ReportPrintingService\_Dest
ECHO    Clean [FileType %FileType%] [Age %RetentionDays%] [DirPath %dirTarget%] --
FORFILES /P %dirTarget% /M %FileType% /S /C "%cmdAction%" /D %RetentionDays%
ECHO    --- ---
REM:-----

REM:-----Direct Flow-----
IF %Environment%=="DevTest" GOTO DoDevTest
IF %Environment%=="PROD" GOTO DoPROD
REM:-----

:DoDevTest
REM:-----DoDevTest-----
ECHO    ---DoDevTest---
REM:--ClinicalDEV--
SET Instance=ClinicalDEV
ECHO    ~~~~~Instance Cleanup = %Instance%~~~~~
CALL "%dirBatch%\StarlimsCleanUp.bat" %Instance% %TestMode% %RetentionDays%
CALL "%dirBatch%\iisCleanUp.bat" %Instance% %TestMode% %RetentionDays%
ECHO    ~~~~~END Instance Cleanup ~~~~~
ECHO.
REM:-- --
REM:--ClinicalTEST--
SET Instance=ClinicalTEST
ECHO    ~~~~~Instance Cleanup = %Instance%~~~~~
CALL "%dirBatch%\StarlimsCleanUp.bat" %Instance% %TestMode% %RetentionDays%
CALL "%dirBatch%\iisCleanUp.bat" %Instance% %TestMode% %RetentionDays%
ECHO    ~~~~~END Instance Cleanup ~~~~~
ECHO.
REM:-- --
REM:--ChemistryDEV--
SET Instance=ChemistryDEV
ECHO    ~~~~~Instance Cleanup = %Instance%~~~~~
CALL "%dirBatch%\StarlimsCleanUp.bat" %Instance% %TestMode% %RetentionDays%
CALL "%dirBatch%\iisCleanUp.bat" %Instance% %TestMode% %RetentionDays%
ECHO    ~~~~~END Instance Cleanup ~~~~~
ECHO.
REM:-- --
```

Continued on Next Page



STARLiMS Clean Batch Specifications

```
REM:--ChemistryTEST--
SET Instance=ChemistryTEST
ECHO      ~~~~~Instance Cleanup = %Instance%~~~~~
CALL "%dirBatch%\StarlimsCleanUp.bat" %Instance% %TestMode% %RetentionDays%
CALL "%dirBatch%\iisCleanUp.bat" %Instance% %TestMode% %RetentionDays%
ECHO      ~~~~~END Instance Cleanup ~~~~~
ECHO.
REM:-- --
ECHO  --- ---
GOTO iisReset
REM:-----

:DoPROD
REM:-----DoPROD-----
ECHO  ---DoPROD---
REM:--ClinicalPROD--
SET Instance=ClinicalPROD
ECHO      ~~~~~Instance Cleanup = %Instance%~~~~~
CALL StarlimsCleanUp %Instance% %TestMode% %RetentionDays%
CALL iisCleanUp %Instance% %TestMode% %RetentionDays%
ECHO      ~~~~~END Instance Cleanup ~~~~~
ECHO.
REM:-- --
REM:--ChemistryPROD--
SET Instance=ChemistryPROD
ECHO      ~~~~~Instance Cleanup = %Instance%~~~~~
CALL StarlimsCleanUp %Instance% %TestMode% %RetentionDays%
CALL iisCleanUp %Instance% %TestMode% %RetentionDays%
ECHO      ~~~~~END Instance Cleanup ~~~~~
ECHO.
REM:-- --
ECHO  --- ---
GOTO iisReset
REM:-----

:iisReset
REM:-----iisReset-----
ECHO  ---iisReset---
IF %TestMode% NEQ "Debug" iisreset
ECHO  --- ---
GOTO TheEnd
REM:-----

:NothingToDo
REM:-----NothingToDo-----
ECHO  ---NothingToDo---
ECHO  No Parameters Passed, Nothing to do
ECHO  --- ---
GOTO TheEnd
REM:-----

:TheEnd
REM:-----TheEnd-----
ECHO  ---TheEnd of MainCleanUP---
ECHO  Time: %time%
ECHO  Exiting Script
ECHO  --- ---
REM:-----

ECHO  -----
REM: -----
```



STARLiMS Clean Batch Specifications

StarlimsCleanUp.bat

```
ECHO OFF
REM: -----StarlimsCleanUp-----
  REM:---DESCRIPTION---
    REM: This script will clean up files associated with the STARLiMS application
    REM: that need to be purged periodically.
  REM:--- ---

  REM:---PARAMETERS---
    REM: 1. Instance
      REM: ClinicalDEV, ClinicalTEST, ClinicalPROD,
      REM: ChemistryDEV, ChemistryTEST, ChemistryPROD
    REM: 2. TestMode
      REM: CleanHouse, Debug
    REM: 3. RetentionDays
  REM:--- ---

  REM:---HISTORY---
    REM: 2018.11.16 JTY: Create bat file
  REM:--- ---
REM: -----
REM: -----

REM:-----Script Setup-----
CLS
SET Instance="%~1"
SET TestMode="%~2"
SET RetentionDays=%3
IF %RetentionDays% gtr 0 (
  SET RetentionDays=-%RetentionDays%
)
ECHO -----StarlimsCleanUp-----
ECHO ---StartUp---
ECHO   Date: %date%
ECHO   Time: %time%
ECHO   Instance = %Instance%
ECHO   TestMode = %TestMode%
ECHO   RetentionDays = %RetentionDays%
ECHO   --- ---
  IF %Instance%==" " GOTO NothingToDo
REM:-----

REM:-----TestMode-----
  IF %TestMode%=="Debug" (
    ECHO ON
    SET cmdAction=cmd /c echo Kill @file
  )
  IF %TestMode%=="CleanHouse" (SET cmdAction=cmd /c del @path /Q)
REM:-----
```

Continued on Next Page



STARLiMS Clean Batch Specifications

```
REM:-----CleanDirectories-----
SET FileType=*.log
IF %Instance%=="ClinicalDEV" SET appPoolDir=STARLIMS10.Clinical107-DEV
IF %Instance%=="ClinicalTEST" SET appPoolDir=STARLIMS10.Test
IF %Instance%=="ClinicalPROD" SET appPoolDir=STARLIMS10.ClinicalLims.azdhs.gov
IF %Instance%=="ChemistryDEV" SET appPoolDir=STARLIMS10.ChemLims-Dev.azdhs.gov
IF %Instance%=="ChemistryTEST" SET appPoolDir=STARLIMS10.ChemLims-Test.azdhs.gov
IF %Instance%=="ChemistryPROD" SET appPoolDir=STARLIMS10.ChemLims.azdhs.gov

REM:--Clean Log Dir--
SET dirTarget=C:\STARLIMS\%appPoolDir%\Log
ECHO      Clean [FileType %FileType%] [Age %RetentionDays%] [DirPath %dirTarget%] --
FORFILES /P %dirTarget% /M %FileType% /S /C "%cmdAction%" /D %RetentionDays%

REM:--Clean UserLog Dir--
SET FileType= *.log
SET dirTarget=C:\STARLIMS\%appPoolDir%\Log\Users
ECHO      Clean [FileType %FileType%] [Age %RetentionDays%] [DirPath %dirTarget%] --
FORFILES /P %dirTarget% /M %FileType% /S /C "%cmdAction%" /D %RetentionDays%

REM:--Clean ReportsCache Dir--
SET FileType= *
SET dirTarget=C:\STARLIMS\%appPoolDir%\WorkPath\ReportsCache
ECHO      Clean [FileType %FileType%] [Age %RetentionDays%] [DirPath %dirTarget%] --
FORFILES /P %dirTarget% /M %FileType% /S /C "%cmdAction%" /D %RetentionDays%
REM: Also removing any sub-directories in this folder.
FOR /D %p IN ("%dirTarget%\*") DO RMDIR /Q /S "%p"

REM:--Clean ShadowCache Dir--
SET FileType= *
SET dirTarget=C:\STARLIMS\%appPoolDir%\ShadowCache
ECHO      Clean [FileType %FileType%] [Age %RetentionDays%] [DirPath %dirTarget%] --
FORFILES /P %dirTarget% /M %FileType% /S /C "%cmdAction%" /D %RetentionDays%
REM: Also removing any sub-directories in this folder.
FOR /D %p IN ("%dirTarget%\*") DO RMDIR /Q /S "%p"

REM:--Clean WorkPath Temp Dir--
SET FileType= *
SET dirTarget=C:\STARLIMS\%appPoolDir%\WorkPath\Temp
ECHO      Clean [FileType %FileType%] [Age %RetentionDays%] [DirPath %dirTarget%] --
DEL "%dirTarget%\*" /F /Q
REM: Also removing any sub-directories in this folder.
FOR /D %p IN ("%dirTarget%\*") DO RMDIR /Q /S "%p"
ECHO      ---

GOTO TheEnd
REM:-----
```

Continued on Next Page



STARLiMS Clean Batch Specifications

```
:NothingToDo
REM:-----NothingToDo-----
ECHO  ---NothingToDo---
ECHO   No Parameters Passed, Nothing to do
ECHO  --- ---
GOTO TheEnd
REM:----- -----

:TheEnd
REM:-----TheEnd-----
ECHO  ---TheEnd of StarlimsCleanUP---
ECHO   Time: %time%
ECHO   Exiting Script
ECHO  --- ---
REM:----- -----
ECHO  -----
REM: -----
```

[Back to Index](#)



STARLiMS Clean Batch Specifications

iisCleanUp.bat

```
ECHO OFF
REM: -----iisCleanUp-----
REM:---DESCRIPTION---
REM: This script will take the passed Instance, and perform a cleanup
REM: of the IIS cache files. Accomplished by stopping the Application Pool
REM: and cleaning the appropriate directories and then starting the
REM: Application Pool
REM:--- ---

REM:---PARAMETERS---
REM: 1. Instance
REM: ClinicalDEV, ClinicalTEST, ClinicalPROD,
REM: ChemistryDEV, ChemistryTEST, ChemistryPROD
REM: 2. TestMode
REM: CleanHouse, Debug
REM:--- ---

REM:---HISTORY---
REM: 2018.11.16 JTY: Create bat file
REM:--- ---
REM: -----
REM: -----

REM:-----Script Setup-----
CLS
SET Instance="%~1"
SET TestMode="%~2"
ECHO -----iisCleanUp-----
ECHO ---StartUp---
ECHO Date: %date%
ECHO Time: %time%
ECHO Instance = %Instance%
ECHO TestMode = %TestMode%
ECHO --- ---
IF %Instance%==" " GOTO NothingToDo
REM:-----

REM:-----TestMode-----
IF %TestMode%=="Debug" (
ECHO ON
SET cmdAction=cmd /c echo Kill @file
)
IF %TestMode%=="CleanHouse" (SET cmdAction=cmd /c del @path /Q)
REM:-----

REM:-----Clean IIS Directories-----
IF %Instance%=="ClinicalDEV" SET AppPoolName=STARLiMS10.Clinical107-DEV
IF %Instance%=="ClinicalTEST" SET AppPoolName=STARLiMS10.Test
IF %Instance%=="ClinicalPROD" SET AppPoolName=STARLiMS10.ClinicalLims.azdhs.gov
IF %Instance%=="ChemistryDEV" SET AppPoolName=STARLiMS10.ChemLims-Dev.azdhs.gov
IF %Instance%=="ChemistryTEST" SET AppPoolName=STARLiMS10.ChemLims-Test.azdhs.gov
IF %Instance%=="ChemistryPROD" SET AppPoolName=STARLiMS10.ChemLims.azdhs.gov
SET appPoolDir=%appPoolName%

ECHO ---Clean IIS Directories---
```

Continued on Next Page



STARLiMS Clean Batch Specifications

```
IF %AppPoolName%==" " GOTO NothingToDo
SET dirPoolLocalCache=C:\STARLiMS\%AppPoolName%\bin\LocalCache
ECHO     AppPoolName = %AppPoolName%
ECHO     dirPoolLocalCache = %dirPoolLocalCache%
SET dirInet=C:\Windows\System32\inetmgr
SET FileType=*
SET dirTarget=%dirPoolLocalCache%
ECHO     Instance %Instance% AppPool and LocalCache cleanup---
IF %TestMode% NEQ "Debug" (
    %dirInet%\appcmd stop apppool /apppool.name:%AppPoolName%
DEL "%dirTarget%\*" /F /Q
REM: Clean Sub-directories
FOR /D %%p IN ("%dirTarget%\*") DO RMDIR /Q /S "%p"
    %dirInet%\appcmd start apppool /apppool.name:%AppPoolName%
)
ECHO     ---
GOTO TheEnd
REM:-----

:NothingToDo
REM:-----NothingToDo-----
ECHO     ---NothingToDo---
ECHO     No Parameters Passed, Nothing to do
ECHO     ---
GOTO TheEnd
REM:-----

:TheEnd
REM:-----TheEnd-----
ECHO     ---TheEnd of iisCleanUp---
ECHO     Time: %time%
ECHO     Exiting Script
ECHO     ---
REM:-----

ECHO     -----

REM: -----
```

[Back to Index](#)